
Preface

If you are an experienced SAS user, you probably already know about ODS Graphics (also called ODS
Statistical Graphics) and how to do the following:

� enable ODS Graphics so that analytical procedures produce graphical output together with tabular
output

� modify graph templates to modify the output that analytical procedures produce
� work with SAS styles
� use SG procedures such as PROC SGPLOT to produce custom plots

If these topics are not already familiar to you, then you should start by looking at the chapters and book
listed in the section “About the Author” on page v. If you are familiar with these topics and want to learn
more, then you have found the right book. This book goes beyond the basics and works through a series
of advanced examples. Each example illustrates a small fraction of the incredible power of ODS Graphics.
In particular, there are many examples of PROC SGPLOT and the SG annotation facility. The last chapter
shows you how to customize every aspect of the graphs that analytical procedures produce.

Sample Code

The beginning of every example has a link that you can double click to see the SAS code for that example.
The �rst example has the following link:

These links work in Adobe Acrobat Reader and Internet Explorer but not in Google Chrome or Mozilla
Firefox.

About the Author

This book was written by Warren F. Kuhfeld, Distinguished Research Statistician Developer in SAS/STAT
Research and Development. Warren received his PhD in psychometrics from the University of North Carolina
at Chapel Hill in 1985 and joined SAS in 1987. He has used SAS since 1979 and has developed SAS
procedures since 1984.

In addition to writing the SAS Press bookStatistical Graphics in SAS: An Introduction to the Graph
Template Language and the Statistical Graphics Procedures, Warren also wrote the following SAS/STAT
documentation chapters: Chapter 20, “Using the Output Delivery System” (SAS/STAT User's Guide),
Chapter 21, “Statistical Graphics Using ODS” (SAS/STAT User's Guide), Chapter 22, “ODS Graphics
Template Modi�cation” (SAS/STAT User's Guide), Chapter 23, “Customizing the Kaplan-Meier Survival
Plot” (SAS/STAT User's Guide).

Warren maintains 11 SAS/STAT procedures. He developed 20 SAS macros for experimental designs for linear
and choice models. His 1,309-page book about discrete choice and other marketing research methods,Market-
ing Research Methods in SAS, is free on the web:http://support.sas.com/resources/papers/
tnote/tnote_marketresearch.html . Warren has also developed the world's largest collection of
orthogonal arrays:http://support.sas.com/techsup/technote/ts723.html .

Chapter 1

Introduction

This book discusses a series of examples that use ODS Graphics. A few examples use the graph template
language (GTL), most use the SGPLOT procedure, and many use statistical graphics (SG) annotation. The
�rst example (see the section “Multiple Axes, Offsets, and Drop Lines” on page 3) introduces multiple axes
and offsets, which are used throughout many of the other examples. This book also discusses axis tables
(Chapter 3, “Axis Tables,”), scatter plot label placement (Chapter 6, “Plots of Labeled Points,”), forest plots
(see the section “Creating a Forest Plot Using PROC SGPLOT” on page 42), adverse event plots (see the
section “Adverse Events Plot” on page 145), dynamic variable modi�cation (see the section “Changing
Dynamic Variables by Using the ODS Document” on page 197), attribute maps (see the section “Attribute
Maps” on page 148), and many other topics. The last three examples (see the sections “Changing Dynamic
Variables by Using the ODS Document” on page 197, “Annotating Single-Panel Graphs That Analytical
Procedures Produce” on page 217, and “Annotating Multiple-Panel Graphs That Analytical Procedures
Produce” on page 228) show you how to modify every aspect of the graphs that analytical procedures produce
and even annotate the graphs. For a list of all SG annotation functions and variables, see the section “SG
Annotation Functions, Variables, and Their Values” on page 133.

Earlier examples explain the code in great detail. However, even the earlier examples assume that you are
already familiar with ODS Graphics; they do not explain all the basics (such as styles, paths, templates, item
stores, destinations, attribute priority, and so on). The functions of many options are obvious from their
names, so they are often not explained. Later examples mostly explain newly introduced features and options.

For more information about ODS Graphics, see Chapter 21, “Statistical Graphics Using ODS” (SAS/STAT
User's Guide), and Chapter 22, “ODS Graphics Template Modi�cation” (SAS/STAT User's Guide). ODS
Graphics documentation includes:SAS ODS Graphics: Procedures Guide, SAS Graph Template Language:
User's Guide, andSAS Graph Template Language: Reference. SAS Press books include Kuhfeld (2010) and
Matange and Heath (2011).

References

Kuhfeld, W. F. (2010).Statistical Graphics in SAS: An Introduction to the Graph Template Language and the
Statistical Graphics Procedures. Cary, NC: SAS Institute Inc.

Matange, S., and Heath, D. (2011).Statistical Graphics Procedures by Example: Effective Graphs Using
SAS. Cary, NC: SAS Institute Inc.

Chapter 2

Axes

Contents
Multiple Axes, Offsets, and Drop Lines . . . . . . . . . . . . . . . . . . . . . . . . . . . . 3
Multiple Axes and Highlighted Points . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 13
Multiple Axes, Axis Alignment, and Many Tick Labels . . . . . . . . . . . . . . . . . . . . 17
Broken Axes . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 24
Multiple Plots with Equated Axes . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . . 27

A graph can have up to four axes: Y on the left, Y2 on the right, X on the bottom, and X2 on the top.
Typically, most graphs have two axes (Y on the left and X on the bottom) and all points have coordinates on
these two axes. Additional axes enable you to display more information in a single graph than you could
reasonably display by using only X and Y axes. Multiple axes and offsets are heavily used in subsequent
examples. These techniques, while simple, provide powerful tools for making advanced graphs.

Multiple Axes, Offsets, and Drop Lines

This example uses a small number of arti�cial data points to illustrate multiple axes and offsets. This example
also illustrates drop lines, which are used to show the correspondence between points and axes.

The following steps create and display the data in Figure 2.1:

data x(drop=i);
do i = 1 to 10 by 3;

x = i + uniform(104);
y = 11 - x + uniform(104);

x2 = i + 10 + uniform(104);
y2 = x2 + uniform(104);

x3 = i + uniform(104);
y3 = 15.5 + uniform(104);

x4 = 15.5 + uniform(104);
y4 = x3 + uniform(104);
output;

end;
run;

proc sgplot noautolegend;
title �Four Groups of Observations�;
scatter y=y x=x / markerattrs=(symbol=diamondfilled);
scatter y=y2 x=x2 / markerattrs=(symbol=circlefilled);
scatter y=y3 x=x3 / markerattrs=(symbol=squarefilled);
scatter y=y4 x=x4 / markerattrs=(symbol=trianglefilled);

run;

Figure 2.1 Four Groups of Observations

The NOAUTOLEGEND option in the PROC SGPLOT statement suppresses the legend. The four SCATTER
statements make four graphs, all in the same set of X and Y axes, and each statement explicitly controls
the marker attributes. PROC SGPLOT automatically cycles the unspeci�ed attributes across statements,
selecting attributes from theGraphData n style elements. Hence, the �rst points, the �lled diamonds, are
blue (which comes from theGraphData1 style element). Continuing, the circles are red (GraphData2 ), the
squares are green (GraphData3 ), and the triangles are brown (GraphData4 ). Almost every graph in this
book is constructed by using the HTMLBlue style. Other styles use other colors. You can disable automatic
attribute cycling by specifying the NOCYCLEATTRS statement in the PROC SGPLOT statement.

The axis labels correspond to the name of the �rst variable that is plotted on each axis. All graphs in this
example are constructed so that the variable names that are displayed as axis labels also identify the axis type:
X, X2, Y, and Y2. The �rst few parts of this example begin by using a subset of these data (the diamonds and
circles). The remainder of the example uses all four groups.

All of the following steps in this example display data and drop lines that run from each point to the axes that
are used for that point. The following macro controls the appearance of the drop lines:

%macro drop(n);
dropto=both lineattrs=graphdata&n(pattern=3) transparency=0.5

%mend;

A macro is not needed; it is used to minimize typing. The DROPTO=BOTH option extends lines to
both axes. The PATTERN=3 option speci�es a dashed line, and TRANSPARENCY=0.5 makes the line
partially transparent. The colors of the drop lines are determined by the style elementsGraphData1 through
GraphData4 , where the index of theGraphData n style element is controlled by the macro variable&n. The
following step creates Figure 2.2:

proc sgplot noautolegend;
title �Drop Lines Show That Both Groups Use the X and Y Axes�;
scatter y=y x=x / markerattrs=(symbol=diamondfilled);
scatter y=y2 x=x2 / markerattrs=(symbol=circlefilled);
dropline y=y x=x / %drop(1);
dropline y=y2 x=x2 / %drop(2);

run;

Figure 2.2 Simple Drop Lines

Like most scatter plots, this graph has an X axis and a Y axis. For both sets of points, the drop lines extend
downward to the X axis and to the left to the Y axis.

The following step creates Figure 2.3:

proc sgplot noautolegend;
title �Blue Diamonds Use the Y Axis, Red Circles Use the Y2 Axis�;
scatter y=y x=x / markerattrs=(symbol=diamondfilled);
scatter y=y2 x=x2 / y2axis markerattrs=(symbol=circlefilled);
dropline y=y x=x / %drop(1);
dropline y=y2 x=x2 / y2axis %drop(2);

run;

Figure 2.3 Two Y Axes

This graph has X, Y, and Y2 axes. The Y2 axis is used when the Y2AXIS option is speci�ed. The X2 axis is
used when the X2AXIS option is speci�ed (which is �rst shown in the next step). X and Y axes are used
when no other axis options are speci�ed. The drop lines for the blue diamonds extend to the left to the Y axis
and downward to the X axis, which is the default behavior when neither the X2AXIS nor the Y2AXIS option
is speci�ed. Because both the second SCATTER and the second DROPLINE statements contain the Y2AXIS
option, the drop lines for the red circles extend downward to the X axis and to the right to the Y2 axis. The
range of values for the Y axis is independent of the range of values for the Y2 axis.

The following step creates Figure 2.4:

proc sgplot noautolegend;
title �Blue Diamonds Use the X Axis, Red Circles Use the X2 Axis�;
scatter y=y x=x / markerattrs=(symbol=diamondfilled);
scatter y=y2 x=x2 / x2axis markerattrs=(symbol=circlefilled);
dropline y=y x=x / %drop(1);
dropline y=y2 x=x2 / x2axis %drop(2);

run;

Figure 2.4 Two X Axes

This graph has X, X2, and Y axes. The drop lines for the blue diamonds extend downward to the X axis and
to the left to the Y axis. The drop lines for the red circles extend upward to the X2 axis and to the left to the
Y axis. The range of values for the X axis is independent of the range of values for the X2 axis.

The following step creates Figure 2.5:

proc sgplot noautolegend;
title �Blue Diamonds Use the X and Y Axes, �

�Red Circles Use the X2 and Y2 Axes�;
scatter y=y x=x / markerattrs=(symbol=diamondfilled);
scatter y=y2 x=x2 / x2axis y2axis markerattrs=(symbol=circlefilled);
dropline y=y x=x / %drop(1);
dropline y=y2 x=x2 / x2axis y2axis %drop(2);

run;

Figure 2.5 Two Y Axes and Two X Axes

This graph has X, X2, Y, and Y2 axes. The �rst scatter plot uses the X and Y axes, and the second scatter
plot uses the X2 and Y2 axes. The drop lines for the blue diamonds extend downward to the X axis and to
the left to the Y axis. The drop lines for the red circles extend to the upward to the X2 axis and to the right to
the Y2 axis. The range of values for the X axis is independent of the range of values for the X2 axis, and the
range of values for the Y axis is independent of the range of values for the Y2 axis.

The following step creates Figure 2.6:

proc sgplot noautolegend;
title �Four Groups of Observations�;
scatter y=y x=x / markerattrs=(symbol=diamondfilled);
scatter y=y2 x=x2 / markerattrs=(symbol=circlefilled);
scatter y=y3 x=x3 / markerattrs=(symbol=squarefilled);
scatter y=y4 x=x4 / markerattrs=(symbol=trianglefilled);
dropline y=y x=x / %drop(1);
dropline y=y2 x=x2 / %drop(2);
dropline y=y3 x=x3 / %drop(3);
dropline y=y4 x=x4 / %drop(4);

run;

Figure 2.6 Four Groups

In this and subsequent steps, all four groups of data are displayed. This graph has X and Y axes. The drop
lines for all points extend downward to the X axis and to the left to the Y axis.

The following step creates Figure 2.7:

proc sgplot noautolegend;
title h=1 �Blue Diamonds Use the X and Y Axes, �

�Red Circles Use the X2 and Y2 Axes,�;
title2 h=1 �Brown Triangles Use the X2 and Y Axes, �

�Green Squares Use the X and Y2 Axes�;
scatter y=y x=x / markerattrs=(symbol=diamondfilled);
scatter y=y2 x=x2 / x2axis y2axis markerattrs=(symbol=circlefilled);
scatter y=y3 x=x3 / y2axis markerattrs=(symbol=squarefilled);
scatter y=y4 x=x4 / x2axis markerattrs=(symbol=trianglefilled);
dropline y=y x=x / %drop(1);
dropline y=y2 x=x2 / x2axis y2axis %drop(2);
dropline y=y3 x=x3 / y2axis %drop(3);
dropline y=y4 x=x4 / x2axis %drop(4);

run;

Figure 2.7 All Four Axes

This graph has X, X2, Y, and Y2 axes. The drop lines for the blue diamonds extend downward to the X axis
and to the left to the Y axis. The drop lines for the red circles extend upward to the X2 axis and to the right to
the Y2 axis. The drop lines for the brown triangles extend upward to the X2 axis and to the left to the Y axis.
The drop lines for the green squares extend downward to the X axis and to the right to the Y2 axis. The range
of values for the X axis is independent of the range of values for the X2 axis, and the range of values for the
Y axis is independent of the range of values for the Y2 axis.

The following step creates Figure 2.8:

proc sgplot noautolegend;
title �Axes Are Extended�;
title2 �Blue Diamonds Use the X and Y Axes, �

�Red Circles Use the X2 and Y2 Axes�;
title3 �Brown Triangles Use the X2 and Y Axes, �

�Green Squares Use the X and Y2 Axes�;
scatter y=y x=x / markerattrs=(symbol=diamondfilled);
scatter y=y2 x=x2 / x2axis y2axis markerattrs=(symbol=circlefilled);
scatter y=y3 x=x3 / y2axis markerattrs=(symbol=squarefilled);
scatter y=y4 x=x4 / x2axis markerattrs=(symbol=trianglefilled);
dropline y=y x=x / %drop(1);
dropline y=y2 x=x2 / x2axis y2axis %drop(2);
dropline y=y3 x=x3 / y2axis %drop(3);
dropline y=y4 x=x4 / x2axis %drop(4);
xaxis min=0 max=20;
yaxis min=0 max=20;
x2axis min=0 max=20;
y2axis min=0 max=20;

run;

This graph has X, X2, Y, and Y2 axes. All four axes are extended so that the ranges for the X and X2 axes
match each other and the ranges of the Y and Y2 axes also match each other. The X, X2, Y, and Y2 axes are
controlled by the statements XAXIS, X2AXIS, YAXIS, and Y2AXIS, respectively; the MIN= and MAX=
options control the ranges.

Figure 2.8 All Four Axes Extended

The drop lines for the blue diamonds extend downward to the X axis and to the left to the Y axis. The drop
lines for the red circles extend upward to the X2 axis and to the right to the Y2 axis. The drop lines for the
brown triangles extend upward to the X2 axis and to the left to the Y axis. The drop lines for the green
squares extend downward to the X axis and to the right to the Y2 axis.

The following step creates Figure 2.9:

proc sgplot noautolegend;
title �Offsets Separate the Four Groups�;
title2 �Blue Diamonds Use the X and Y Axes, �

�Red Circles Use the X2 and Y2 Axes,�;
title3 �Brown Triangles Use the X2 and Y Axes, �

�Green Squares Use the X and Y2 Axes�;
scatter y=y x=x / markerattrs=(symbol=diamondfilled);
scatter y=y2 x=x2 / x2axis y2axis markerattrs=(symbol=circlefilled);
scatter y=y3 x=x3 / y2axis markerattrs=(symbol=squarefilled);
scatter y=y4 x=x4 / x2axis markerattrs=(symbol=trianglefilled);
dropline y=y x=x / %drop(1);
dropline y=y2 x=x2 / x2axis y2axis %drop(2);
dropline y=y3 x=x3 / y2axis %drop(3);
dropline y=y4 x=x4 / x2axis %drop(4);
xaxis offsetmax=.52;
yaxis offsetmax=.52;
x2axis offsetmin=.52;
y2axis offsetmin=.52;

run;

This graph has X, X2, Y, and Y2 axes, and each axis has an offset. The OFFSETMAX= option reserves space
to the right or above, and the OFFSETMIN= option reserves space to the left or below. The blue diamonds
are plotted in the bottom left, and space is reserved to the right and above for other parts of the graph. The
red circles are plotted in the top right, and space is reserved to the left and below for other parts of the graph.
The brown triangles are plotted in the bottom right, and space is reserved to the left and above for other parts
of the graph. The green squares are plotted in the top left, and space is reserved to the right and below for
other parts of the graph.

Figure 2.9 Offsets Separate the Four Groups

The drop lines for the blue diamonds extend downward to the X axis and to the left to the Y axis. The drop
lines for the red circles extend upward to the X2 axis and to the right to the Y2 axis. The drop lines for the
brown triangles extend upward to the X2 axis and to the left to the Y axis. The drop lines for the green
squares extend downward to the X axis and to the right to the Y2 axis. The range of values for the X axis is
independent of the range of values for the X2 axis, and the range of values for the Y axis is independent of
the range of values for the Y2 axis.

This approach (multiple axes and offsets that reserve space for other parts of the graph) is extensively used in
subsequent examples. It is important to understand the techniques illustrated in this example before going on
to subsequent examples.

Multiple Axes and Highlighted Points

This example shows how to construct a graph that has multiple axes, and it highlights selected points in
the graph. The data are from the mileage records of someone training to run a marathon (26.2 miles). The
following step reads the data:

%let start = �01jan2013�d;
%let stop = �29apr2014�d;
data runs(keep=date distance totalmiles);

retain year � �;
input d $ Distance @@;
if nmiss(distance) then do; year = �/� || d; return; end;
Date = input(trim(d) || year, mmddyy10.);
TotalMiles + distance;
if date ge &stop - 364 then PastYear + distance;
call symputx(�t�, put(pastyear, 4.));
output;

datalines;
2013 . 1/1 7.6 1/3 7.6 1/5 10.4 1/7 7.6 1/9 7.6 1/13 10.4 1/24 7.6 1/28 10.4
1/30 7.6 2/1 7.6 2/3 12.2 2/5 7.6 2/10 7.6 2/12 8.2 2/14 8.2 2/16 13.1 2/19 8.2
2/21 8.2 2/24 10.4 2/26 7.6 2/28 8.2 3/2 13.65 3/4 8.2 3/6 8.2 3/8 13.9 3/10
10.4 3/12 8.2 3/14 8.2 3/16 8.2 3/17 5.2 3/19 8.2 3/21 8.2 3/23 8.2 3/27 8.2
3/28 5.2 3/30 14 4/1 5.2 4/3 8.2 4/5 5.2 4/6 10.4 4/8 8.2 4/10 8.2 4/11 5.2 4/13
10.4 4/15 8.2 4/17 5.2 4/20 8.2 4/22 8.2 4/24 8.2 4/26 8.2 4/27 3 4/28 4 4/29
4.25 4/30 5.2 5/1 4.1 5/2 4.1 5/4 8.2 5/6 5.2 5/7 8.2 5/9 5.2 5/11 10.4 5/13
4.35 5/14 4.25 5/15 4.25 5/22 5.2 5/25 10.4 5/27 10.4 5/29 10.4 5/31 14 6/2 10.4
6/4 10.4 6/6 10.4 6/8 10.4 6/9 10.4 6/11 10.4 6/15 8.2 6/19 5.2 6/23 10.4 6/25
8.2 6/26 5.2 6/28 15 6/30 11.2 7/2 3 7/3 8.2 7/5 8.2 7/6 8.2 7/7 8.2 7/9 8.2
7/10 8.2 7/12 16 7/15 8.2 7/16 8.2 7/17 5.2 7/19 14 7/20 8.2 7/22 8.2 7/23 5.2
7/24 10.4 7/26 8.2 7/28 4.1 7/29 6.3 7/30 10.4 7/31 8.2 8/3 11.9 8/4 4.5 8/5 5
8/6 6.1 8/7 4.2 8/9 5.2 8/10 4 8/11 12 8/13 5.2 8/16 5.2 8/17 10.4 8/18 8.2 8/19
5.2 8/20 5.2 8/21 8.2 8/23 5.4 8/27 5.2 8/28 5.2 8/29 3 9/1 5.2 9/3 5.2 9/4 8.2
9/5 10.4 9/6 5.2 9/7 5.2 9/8 8.9 9/9 7 9/12 5.2 9/14 18 9/17 5.2 9/18 10.4 9/19
5.2 9/20 6.75 9/21 10.4 9/23 8.2 9/24 8.2 9/27 8.2 9/28 20 10/2 5.2 10/3 8.2
10/4 5.2 10/5 14 10/8 9.8 10/10 10 10/11 5.2 10/13 20 10/15 8.2 10/16 5.2 10/17
8.2 10/22 21.3 10/24 8 10/25 5.2 10/26 14 10/30 5.2 10/31 5.2 11/1 5.2 11/2 10.4
11/5 3.1 11/7 3.1 11/9 27.36 11/11 5.2 11/12 5.2 11/13 5.2 11/15 14 11/16 8.2

11/17 5.2 11/18 5.2 11/20 10.4 11/21 5.24 11/24 10.4 11/25 8.2 11/26 5.2 11/28
10.4 12/2 5.2 12/3 8.2 12/4 10.4 12/5 5.2 12/6 10.4 12/7 10.4 12/8 5.2 12/9 8.3
12/11 10.5 12/12 8.2 12/13 5.3 12/14 10.4 12/15 8.2 12/16 5.2 12/17 5.3 12/18
10.4 12/19 8.4 12/20 5.2 12/22 8.2 12/23 8.2 12/24 8.2 12/25 8.2 12/26 8.2 12/27
8.2 12/28 5.2 12/30 14 12/31 5.2 2014 . 1/1 5.2 1/2 5.2 1/3 5.2 1/4 5.2 1/5
10.2 1/9 5.2 1/10 5.2 1/11 6.8 1/14 5.2 1/17 5.2 1/18 6.8 1/19 6.8 1/20 5.2 1/21
5.2 1/23 5.2 1/24 5.3 1/25 5.2 1/26 10.4 1/27 7.6 1/28 8.2 1/29 3.3 2/1 10.4 2/2
10.4 2/4 5.2 2/5 8.2 2/6 5.2 2/8 14 2/10 5.2 2/11 5.2 2/12 8.3 2/15 5.2 2/16 8.2
2/18 8.2 2/20 5.2 2/23 18.2 2/25 5.2 2/27 5.2 2/28 21 3/2 8.2 3/4 5.2 3/8 10.4
3/10 3.1 3/12 3.1 3/16 26.3 3/20 3.1 3/21 5.2 3/22 5.2 3/27 5.2 3/28 5.2 3/30 14
3/31 8.2 4/2 5.2 4/4 5.2 4/5 14 4/6 10.4 4/8 5.2 4/10 8.2 4/11 3 4/12 14 4/13
9.1 4/14 5.2 4/16 5.2 4/17 5.2 4/18 14 4/20 14 4/21 5.2 4/22 8.2 4/23 5.2 4/24
5.2 4/25 14 4/27 14 4/29 5.3
;

TheRuns data set has three variables:Distance (distance run that day),Date (date of the run), andTotalMiles
(total miles recorded since the start date). The statementTotalMiles + distance is a SAS sum statement
(which is equivalent toTotalMiles = TotalMiles + distance and also retainsTotalMiles and initializes
it to zero). The following steps prepare the data for graphing:

data all;
Distance = 0;
do Date = &start to &stop;

output;
end;

run;

data r2;
update all runs(in=r);
by date;
Day = ifc(r, �Run �, �Rest�);
if distance ge 26.2 then Star = distance;
x = date - &start;
output;

run;

proc reg data=r2 noprint;
model totalmiles = x / noint;
output p=pred out=r;

run;

The �rst two DATA steps add an observation for each date that the runner did not run. The IFC function
returns�Run� whenr is true (theRuns data set is being processed) and otherwise returns�Rest� . The
PROC REG step �ts the modelTotalMiles = (Date – '01Jan2013'd), where'01Jan2013'd is the starting
date for training. The predicted values,Pred, provide a reference line for the graph of the total miles. The
following step creates the graph displayed in Figure 2.10:

proc sgplot data=r;
title �Marathon Training Data�;
refline %sysevalf(1600 / 365);
series y=pred x=date / y2axis lineattrs=graphreference;

step y=totalmiles x=date/ y2axis lineattrs=graphdata3(thickness=2px)
name=�t� legendlabel=�Total Miles�;

pbspline y=distance x=date / nomarkers name=�p� legendlabel=�Miles Per Day�;
scatter y=distance x=date / group=Day name=�r�

markerattrs=(size=5px symbol=circlefilled);
scatter y=star x=date / markerattrs=(size=15px symbol=star)

legendlabel=�Marathon� name=�s�;
format date mmddyy8.;
xaxis display=(nolabel);
yaxis label=�Daily Run Distance� labelattrs=graphdata1(weight=bold);
y2axis labelattrs=graphdata3 valueattrs=graphdata3 label=�Total Miles�;
keylegend �r� �s� �p� �t� / title=" &t Past Year "

location=inside position=topleft across=1;
run;

Figure 2.10 Graph That Has Two Y Axes

The blue dots show daily runs, and the daily run distance in miles is displayed on the Y axis. The red dots
show resting days.

The horizontal reference line is at approximately 4.4 miles—the average number of miles needed per day to
reach 1,600 miles for the year. The blue curve is a penalized B-spline function that is �t to the distances (all
running and resting days); it shows a smooth function of distance run per day. When the curve is above the

reference line, the runner is exceeding his goal. Notice that the reference line and the penalized B-spline
function would be misleading if the resting days had been excluded from the graph. The graph displays stars
to highlight the marathons (runs of 26.2 miles or longer).

This graph also shows the total miles run over the course of the marathon training. The graph needs two Y
axes because daily runs are on a scale of 0 to 27.36 miles whereas the total distance is on a scale of 0 to more
than 2,204.5 miles. If there were one combined axis, the daily runs would be too compressed to provide
information. Total distance is displayed in the green step function and the Y2 axis (the Y axis on the right).
The diagonal reference line is a regression line that �ts total miles as a function of time. In some periods the
green step function is increasing faster than the average slope, and in other periods it is increasing at a slower
rate.

The graph is constructed as follows:

� The REFLINE statement draws a horizontal reference line at 1,600 miles divided by 365 days and uses
the Y axis.

� The SERIES statement displays the diagonal reference line and uses the Y2 axis.
� The STEP statement displays total miles and uses the Y2 axis.
� The PBSPLINE statement displays a smooth function of distance per day as a function of time and

uses the Y axis.
� The �rst SCATTER statement displays distances as 5-pixel �lled circles and uses the Y axis. The

group variableDay differentiates resting days from running days.
� The second SCATTER statement displays a 15-pixel star for the marathon runs.
� The FORMAT statement declares theDate variable as a date variable. PROC SGPLOT does not

display date variables as literally prescribed by the format. Rather, it extracts the years and labels the
ticks in an elegant way.

� The XAXIS statement suppresses the X axis label. The tick labels are suf�cient to show that the X
axis is a date axis.

� The YAXIS statement sets the Y axis label.
� The Y2AXIS statement uses the right Y axis to display the total miles in the same style as the step plot

(GraphData3 ).
� The KEYLEGEND statement displays a legend that identi�es the points in the plot, the smooth function

of average miles per day, and the total miles.
� Fittingly, a RUN statement ends the step.

The GraphReference style element is used implicitly and explicitly for the two reference lines. The
PBSPLINE �t function uses theGraphFit style element. Running days (the �rst group) are displayed
by using theGraphData1 style element (blue), resting days by usingGraphData2 (red), and total miles
by usingGraphData3 (green). Reference lines are drawn �rst so that they never obscure any other graph
elements. The total miles function is drawn next so that it never obscures distances per day. The PBSPLINE
function is �t to all distances (ignoring groups), whereas the scatter plot of distances displays the running and
resting days as separate groups.

Multiple Axes, Axis Alignment, and Many Tick Labels

The previous example showed how to make a graph that has multiple axes that are independent of each other.
That is the typical usage of multiple axes. This example shows you how to make two axes that have the same
range and same ticks in the same positions, but use different tick labels. The X axis displays the date, and the
X2 axis displays the day of the week. To illustrate, the following step reads a data set that contains datetime
values and a variabley:

data x;
input Time datetime16. y @@;
time2 = time;
datalines;

22AUG15:17:47:59 4 30AUG15:04:46:07 3 05SEP15:23:50:59 4 30AUG15:22:00:02 4
22AUG15:09:59:36 4 28AUG15:11:07:32 5 30AUG15:00:33:31 4 07SEP15:12:40:46 4
23AUG15:04:51:09 4 26AUG15:07:30:54 4 22AUG15:04:29:50 3 07SEP15:21:22:41 4
07SEP15:15:38:58 5 06SEP15:18:10:15 3 08SEP15:14:52:29 4 27AUG15:10:01:01 5
24AUG15:14:49:55 4 04SEP15:09:57:37 4 07SEP15:06:18:21 3 04SEP15:21:26:21 4
25AUG15:23:30:27 3 27AUG15:23:53:22 4 30AUG15:11:14:32 5 30AUG15:06:52:19 4
06SEP15:01:41:58 5 05SEP15:20:03:54 3 04SEP15:20:05:47 3 30AUG15:04:17:00 3
02SEP15:04:33:24 4 23AUG15:19:21:22 4 08SEP15:02:44:52 4 30AUG15:16:46:16 3
22AUG15:17:16:58 4 29AUG15:04:46:29 3 03SEP15:20:07:47 3 23AUG15:15:29:14 5
08SEP15:22:42:49 3 05SEP15:22:53:44 4 22AUG15:06:23:28 5 30AUG15:19:50:41 3
04SEP15:11:07:21 5 22AUG15:09:43:14 5 07SEP15:04:31:59 4 30AUG15:09:43:30 4
23AUG15:13:22:42 6 07SEP15:22:11:57 4 25AUG15:01:36:36 4 04SEP15:03:37:37 4
03SEP15:09:09:53 3 02SEP15:00:57:35 2 27AUG15:20:44:19 4 29AUG15:03:31:12 5
28AUG15:18:05:29 4 31AUG15:18:17:15 4 24AUG15:12:01:32 4 07SEP15:08:43:56 2
03SEP15:13:08:22 3 02SEP15:08:17:34 3 28AUG15:08:09:03 2 24AUG15:15:05:34 3
22AUG15:07:51:43 5 01SEP15:05:52:43 3 01SEP15:04:02:02 4 05SEP15:20:10:14 4
25AUG15:08:49:33 2 06SEP15:03:46:10 3 28AUG15:23:45:55 4 27AUG15:18:56:00 4
23AUG15:02:28:42 4 30AUG15:18:41:04 6 25AUG15:22:52:42 4 31AUG15:04:31:22 4
29AUG15:07:31:17 4 30AUG15:00:04:45 6 25AUG15:21:15:18 6 08SEP15:07:11:50 6
27AUG15:10:42:16 3 30AUG15:18:25:38 4 28AUG15:19:44:59 1 23AUG15:15:37:26 4
25AUG15:13:13:21 5 31AUG15:06:47:43 3 05SEP15:00:51:34 6 07SEP15:13:33:42 5
28AUG15:19:20:32 4 03SEP15:08:08:12 4 09SEP15:05:36:24 3 29AUG15:11:31:29 5
26AUG15:09:41:25 5 30AUG15:16:27:16 4 06SEP15:04:18:38 5 23AUG15:19:51:50 4
29AUG15:07:14:18 6 09SEP15:00:20:35 5 04SEP15:14:47:57 4 25AUG15:13:40:57 5
06SEP15:10:06:47 3 05SEP15:20:13:34 4 25AUG15:06:20:25 4 23AUG15:08:43:42 5
;

This DATA step creates the variableTime (which will be displayed on the X axis) and creates a duplicate
time variableTime2 (which will be used to construct the X2 axis). Although ODS Graphics has sophisticated
options for handling datetime data that work well for a single axis, you will use none of them in this example.
When you assign a date format to an axis, ODS Graphics behaves differently than it behaves for other types
of axes. It can extract the year and display it separately to minimize redundancy in the tick labels (as shown
in Figure 2.10). Furthermore, you lose control over tick label thinning. When there is a single date axis, these
behaviors are generally good. Because this example has two axes and you will force them to correspond, you
will need to do much of the axis labeling work yourself instead of relying on ODS Graphics. You need to
begin by �nding the range of datetime values.

The following step �nds the minimum and maximum datetimes and stores them in a data set:

proc means data=x noprint;
var time;
output min=min max=max out=m;

run;

SAS date variables contain the number of days since January 1, 1960, and SAS datetime variables contain
the number of seconds since January 1, 1960. You can use SAS functions and formats to manipulate date and
datetime variables. The following step manipulates the minimum and maximum datetimes, and creates four
macro variables:

data _null_;
set m;
call symputx(�s� , 24 * 3600);
call symputx(�min� , dhms(datepart(min) , 0, 0, 0));
call symputx(�max� , dhms(datepart(max) + 1, 0, 0, 0));
call symputx(�year�, year(datepart(min)));

run;

The macro variableS contains the number of seconds in a day:24 � 60 � 60. The macro variableMin
contains the SAS datetime value for the beginning of the day that the �rst time occurred. The macro variable
Max contains the SAS datetime value for the beginning of the day after the last time occurred. The macro
variableYear contains the year in which the �rst time occurred.

You can construct a data set and use it to make one or more formats. The following steps create two formats,
WKDAY and WKDATE:

data c;
length l1 l2 $ 12;
retain f1 �wkday� f2 �wkdate� sexcl �N� eexcl �Y� ;
do start = &min to &max by &s;

end = start + &s;
l1 = left(put(datepart(start), downame12.));
l2 = left(put(start, datetime12.));
l2 = substr(l2, 1, 2) || propcase(substr(l2, 3, 3));
output;

end;
run;

proc format cntlin=c(rename=(l1=label f1=fmtname));
run;

proc format cntlin=c(rename=(l2=label f2=fmtname));
run;

This is a one-pass DATA step. There are no input data, so the DATA step ends when it hits the RUN statement.
In contrast, DATA steps that have an INPUT or SET statement usually pass through the step many times
(usually once for every input observation). When there are multiple passes through the DATA step, the
RETAIN statement retains values across passes rather than initializing them to missing each time. The
RETAIN statement provides a convenient syntax for creating and initializing constant variables even when
retaining is not required.

The �rst format displays the values in thel1 variable (days of the week). The values come from writing a
SAS date value by using the DOWNAME (day-of-the-week name) format. The second format displays the
values in thel2 variable (date). The values come from writing a SAS datetime value by using the DATETIME
format and extracting some of the results. The PROPCASE function capitalizes the �rst letter of the month
and sets the remaining two letters to lowercase.

The following steps display the input to the two formats:

proc print noobs data=c(rename=(l1=label f1=fmtname) drop=f2 l2);
format start end datetime16.;

run;

proc print noobs data=c(rename=(l2=label f2=fmtname) drop=f1 l1);
format start end datetime16.;

run;

The FORMAT procedure uses the CNTLIN= option to make a format whose name is extracted from the
FmtName variable. The other variables control what is displayed for each input value. The data in Figure 2.11
contain aLabel variable that contains the values to be displayed (the day of the week) in place of each
datetime range, aStart variable that speci�es the start of the range, anEnd variable that speci�es the end of
the range, ansExcl variable that includes the starting value in the range, and aneExcl variable that excludes
the ending value from the range. Datetime values are displayed asDayMonthYear :Hours:Minutes:Seconds.
The data in Figure 2.11 are similar, but theLabel variable contains the date and theFmtName variable
contains a different format name. In both cases, the DO statement loops from the starting date to the ending
date and creates one observation for every day in the range of days for which there are data.

Figure 2.11 Input to the WKDAY Format De�nition

Figure 2.12 Input to the WKDATE Format De�nition

The following step checks the logic of the code so far to ensure that the X and X2 axes correspond as planned:

proc sgplot data=x noautolegend;
scatter y=y x=time;
scatter y=y x=time2 / x2axis markerattrs=(size=3px);
format time wkdate5. time2 wkday3.;
xaxis grid display=(nolabel noticks) values=(&min to &max by &s)

fitpolicy=rotate;
x2axis display=(nolabel noticks) values=(&min to &max by &s)

fitpolicy=none;
footnote justify=left "&year";

run;

The NOAUTOLEGEND option in the PROC SGPLOT statement suppresses the legend. The �rst SCATTER
statement creates the graph of interest, and it uses the X and Y axes. The second SCATTER statement is used
to check results (for now) and create the X2 axis. For now, the MARKERATTRS=(SIZE=3PX) option in
the second SCATTER statement displays small markers that should occupy the same position as the actual
markers. These markers are removed in the next step. The �rst graph uses theTime variable, and the second
graph uses the identicalTime2 variable. The X and X2 axes both have the same underlying tick values
because both have the same VALUES= option that speci�es ticks that range from the �rst day to the last day
by one-day increments. The X and X2 axes have different tick labels because each X= variable has a different
format. For both axes, tick marks are suppressed because their locations are clear from the grid lines. The
FITPOLICY= option in the X2AXIS statement displays the values on the X2 axis with no thinning, and
the FITPOLICY= option in the XAXIS statement rotates the values on the X axis, again with no thinning.
Thinning refers to the process of removing some ticks for axes that would otherwise become congested. The
year is displayed in the bottom left of the graph.

The results are displayed in Figure 2.13.

Figure 2.13 Check the Logic (Red Markers Should Be inside Blue Markers)

The grid lines show that the ticks align and show the correspondence between the day and date: August
22, 2015, is a Saturday; ...; August 30 is a Sunday; ...; and September 10 is a Thursday. The small red
(Graphdata2 ) markers all appear inside the larger blue (Graphdata1 ) markers, showing that the X and X2
axes are set up correctly.

The following step makes the �nal graph:

proc sgplot data=x noautolegend;
scatter y=y x=time;
scatter y=y x=time2 / x2axis markerattrs=(size=0px);
format time wkdate5. time2 wkday3.;
xaxis grid display=(nolabel noticks) values=(&min to &max by &s)

fitpolicy=rotate;
x2axis display=(nolabel noticks) values=(&min to &max by &s)

fitpolicy=none;
footnote justify=left "&year";

run;

The results are displayed in Figure 2.14.

Figure 2.14 Days on the X2 Axis and Dates on the X Axis

Now there is one set of markers. The second SCATTER statement makes an invisible plot (zero-size markers),
but it populates the X2 axis.

You can make the fonts smaller and display fewer characters on the X2 axis in order to display more data.
The following steps generate some random data and illustrate:

%let s = %eval(24 * 3600);

data x(drop=i);
do i = 1 to 1000;

time = �10SEP2015:15:15:34�dt - 47 * &s + floor(46 * &s * uniform(368));
time2 = time;
y = round(10 * (4 + normal(368)));
output;

end;
run;

proc means data=x noprint;
var time;
output min=min max=max out=m;

run;

data _null_;
set m;
call symputx(�min� , dhms(datepart(min) , 0, 0, 0));
call symputx(�max� , dhms(datepart(max) + 1, 0, 0, 0));
call symputx(�year�, year(datepart(min)));

run;

data c;
length l1 l2 $ 12;
retain f1 �wkday� f2 �wkdate� sexcl �N� eexcl �Y� ;
do start = &min to &max by &s;

end = start + &s;
l1 = left(put(datepart(start), downame12.));
l2 = left(put(start, datetime12.));
l2 = substr(l2, 1, 2) || propcase(substr(l2, 3, 3));
output;

end;
run;

proc format cntlin=c(rename=(l1=label f1=fmtname));
run;

proc format cntlin=c(rename=(l2=label f2=fmtname));
run;

%let o = display=(nolabel noticks) values=(&min to &max by &s)
valueattrs=(size=7px weight=bold);

proc sgplot data=x noautolegend;
scatter y=y x=time / markerattrs=(size=5px);
scatter y=y x=time2 / x2axis markerattrs=(size=0px);
format time wkdate5. time2 wkday2.;
xaxis grid &o fitpolicy=rotate;
x2axis &o fitpolicy=none;
footnote justify=left "&year";

run;

The marker size is decreased in the �rst SCATTER statement. Decreasing the marker size is not required, but
as graphs become more congested, smaller markers often look better. Some axis options are stored in a macro
variable to minimize typing. This too is not required. The VALUEATTRS=(SIZE=7PX WEIGHT=BOLD)
option decreases the size of the tick labels but makes them bold so that they are easier to read. The width of
the WKDAY format is decreased from 3 to 2. The results are displayed in Figure 2.15.

Figure 2.15 Two Aligned X Axes, Many Ticks, and Smaller Tick Labels

Broken Axes

This example illustrates broken axes, which are useful when there are extreme observations that compress
relevant patterns when you use ordinary axes. To illustrate, the following step creates some arti�cial data that
have three groups:

data x(drop=i);
do g = 1 to 3;

do i = 1 to 100;
x = 10 * uniform(7);
y = 10 ** g + log(x) + sin(g * (x + 1));
output;

end;
end;

run;

Because of the expression10 ** g, Y values in the �rst group are of the order 10, Y values in the second
group are of the order 100, and Y values in the third group are of the order 1,000.

The following step creates an ordinary scatter plot:

ods graphics on / attrpriority=none;
proc sgplot data=x noautolegend;

styleattrs datasymbols=(circlefilled squarefilled trianglefilled);
scatter y=y x=x / group=g markerattrs=(size=5px);

run;

You can specify the STYLEATTRS statement along with the ATTRPRIORITY=NONE option to distinguish
groups by colors and markers. The results are displayed in Figure 2.16. Because of the extreme range of
values, the patterns in the three groups are not apparent.

Figure 2.16 Extreme Values Squash the Display of Other Values

You can use PROC MEANS to �nd the minimum and maximum value in each group:

proc means;
var y;
class g;

run;

The results of this step are displayed in Figure 2.17.

Figure 2.17 Minima and Maxima

The results are used in the next step to specify three value ranges, each of which consists of values smaller
than the minimum and larger than the maximum for each group:

proc sgplot data=x noautolegend;
styleattrs datasymbols=(circlefilled squarefilled trianglefilled);
scatter y=y x=x / group=g markerattrs=(size=5px);
yaxis ranges=(8 - 13 98 - 104 995 - 1004);

run;

The results are displayed in Figure 2.18.

Figure 2.18 Broken Axes

Now the three patterns are clear. Also, the broken axis makes it clear that parts of the graph are omitted.
The breaks are slightly curved rather than linear to show that there is one broken axis and not three separate
graphs. Other options for how to break the axes are as follows: AXISBREAK=BRACKET | NOTCH
| SLANTEDLEFT | SLANTEDRIGHT | SQUIGGLE | SPARK | Z. The next step illustrates specifying
AXISBREAK=SPARK in the STYLEATTRS statement:

proc sgplot data=x noautolegend;
styleattrs datasymbols=(circlefilled squarefilled trianglefilled)

axisbreak=spark;
scatter y=y x=x / group=g markerattrs=(size=5px);
yaxis ranges=(8 - 13 98 - 104 995 - 1004);

run;

The results are displayed in Figure 2.19.

Figure 2.19 Broken Axes

Multiple Plots with Equated Axes

This example shows you how to construct a panel that contains multiple plots, each having equated axes (a
centimeter on the Y axis represents the same data range as a centimeter on the X axis). Furthermore, in the
�nal graph, a centimeter on every X axis and Y axis represents the same data range as a centimeter on every
other X axis and Y axis.

This example begins by using PROC CANCORR to output two sets of canonical variables (v1–v6 and
w1–w6):

proc cancorr data=sashelp.baseball(drop=name) out=c(keep=v: w:);
var n:;
with cr:;

run;

You can plot each pair of canonical variables by using PROC SGSCATTER:

proc sgscatter;
plot w1 * v1 w2* v2 w3* v3 w4* v4 w5* v5 w6* v6;

run;

The results are displayed in Figure 2.20

Figure 2.20 Pairwise Plots of Canonical Variables

For this particular application, this graph could be improved. Canonical variables are standardized to mean 0
and variance 1. Because all the variables are on the same scale, it would be nice if all the axes were equated
within and across the plots. It is easy to use the SG procedures to make the overall size of the graph space
(which includes the data area, ticks, axis labels, titles, and footnotes) square and make the ticks the same.
However, for truly equated plots, you must use the GTL. These steps extract the maximum absolute value,t,
from both sets of canonical variables and use it to create a common set of equated axes that extend from –t to
t. The following steps create an intermediate graph that shows the axis ranges:

data _null_;
retain m 0;
set c end=eof;
m = max(m, max(of v: w:), abs(min(of v: w:)));
if eof then call symputx(�t�, m);

run;

%let nplots = 6;
%let nrows = 2;
%let ncols = 3;
%let height = 480px;

proc template;
define statgraph canvars;

begingraph / designheight=&height;
layout lattice / columns=&ncols rows=&nrows

rowgutter=10 columngutter=10;
%macro plot;

%do i = 1 %to &nplots;
layout overlayequated / equatetype=square

commonaxisopts=(viewmin=-&t viewmax=&t
tickvaluelist=(-&t &t));

scatterplot y = w&i x = v&i;
endlayout;

%end;
%mend; %plot

endlayout;
endgraph;

end;
run;

proc sgrender data=c template=canvars;
run;

The DATA _NULL_ step reads the data set of canonical variables, �nds the maximum absolute value across
both sets of variables, and outputs that value to a macro variable. The template is explained in detail when
the �nal version is presented. The results are displayed in Figure 2.21.

Figure 2.21 Pairwise Equated Plots of Canonical Variables

You can see that all graphs are square and all axes range from –6.13 to 6.13. For graphs like these, you do not
need to see the tick or tick labels, but they are shown here to ensure that the results are correct.

The following steps make the �nal graph:

proc template;
define statgraph canvars;

begingraph / designheight=&height;
layout lattice / columns=&ncols rows=&nrows

rowgutter=10 columngutter=10;
%macro plot;

%do i = 1 %to &nplots;
layout overlayequated / equatetype=square

yaxisopts=(display=(label)) xaxisopts=(display=(label))
commonaxisopts=(viewmin=-&t viewmax=&t

tickvaluelist=(-&t &t));
scatterplot y = w&i x = v&i;

endlayout;
%end;

%mend; %plot
endlayout;

endgraph;
end;

run;

proc sgrender data=c template=canvars;
run;

The results are displayed in Figure 2.22.

Figure 2.22 Final Pairwise Equated Plots of Canonical Variables

The template relies on the macro variables and DATA _NULL_ step shown previously. The template creates
a lattice that has three columns and two rows. Row and column gutter space provide separation between
the graphs. The template has six equated overlays (one for each graph), which are constructed by using a
macro to minimize typing. Each graph is speci�ed as EQUATETYPE=SQUARE. The DISPLAY=(LABEL)
option suppresses the ticks and tick labels for each graph, displaying only the axis label and providing more
space for the graphs. The common axis options ensure that all axes extend from –t to t. Specifying the
desired tick value list without specifying the VIEWMIN= and VIEWMAX= options is not suf�cient. The
SCATTERPLOT statement creates the scatter plot.

You could make the code more �exible by �nding the minimum and maximum and using those to scale
the axes. However, for standardized variables such as these, the simpler approach is reasonable. You could
increase the design height in the BEGINGRAPH statement and make3 � 3 or 4 � 4 displays. You could
decrease the design width (not currently speci�ed) in the BEGINGRAPH statement and make2 � 2 displays.

An axis table displays a table of values together with a graph. Examples of the axis table include the inertia
table displayed by PROC CORRESP (see the �rst graph in Figure 3.1), the Kaplan-Meier plot with at-risk
information displayed by PROC LIFETEST (see the second graph in Figure 3.1), and the studentized residuals
and Cook'sD charts displayed by PROC REG (see Figure 3.3 and Figure 3.4). Either the columns of the
table (as in PROC LIFETEST) or the rows (as in PROC CORRESP and PROC REG) align with elements of
the graph.

Figure 3.1 Axis Table Examples

Figure 3.1 continued

Axis Table Example Using PROC REG

This example uses PROC REG to introduce the axis table. When ODS Graphics is enabled and a table of
residuals and other observationwise statistics is requested, PROC REG produces the residual table (shown
in Figure 3.2) and the graphical display of studentized residuals and Cook'sD (shown in Figure 3.3). The
following step illustrates:

ods graphics on;
proc reg data=sashelp.class;

model weight = height / r;
quit;

Figure 3.2 Residuals Table

Figure 3.3 Studentized Residuals and Cook's D

You can use the UNPACK option to display the studentized residuals and Cook'sD in separate charts as
follows:

proc reg data=sashelp.class plots=residualchart(unpack);
model weight = height / r;

quit;

The results are displayed in Figure 3.4.

Figure 3.4 Separate Charts for Studentized Residuals and Cook's D

The graphs of studentized residuals and Cook'sD that are displayed in Figure 3.3 and Figure 3.4 are not
intended to duplicate the table, although the Cook'sD chart in Figure 3.4 comes close. This example shows



Axis Table Example Using PROC REG F 37

how you can add a graph to the table in Figure 3.2. The following step creates an ODS output data setr,
which contains the information in the table:

proc reg data=sashelp.class;
ods output OutputStatistics=r;
model weight = height / r;

quit;

PROC CONTENTS displays the variable names and labels:

proc contents varnum;
ods select position;

run;

The results are displayed in Figure 3.5.

Figure 3.5 Variable Names in the Residuals Table

The following steps create a template and the table and graph in Figure 3.6:

proc template;
define statgraph StudentizedResidualChart;

begingraph; * add: / designheight=height (example: / designheight=2000px);
entrytitle �Residuals�;
layout overlay / walldisplay=none

yaxisopts=(display=none type=discrete reverse=true)
x2axisopts=(label=�Studentized Residuals� displaysecondary=(line)

labelattrs=graphvaluetext(weight=bold)
griddisplay=auto_on linearopts=(integer=true));

innermargin / align=left gutter=5px;
%let o = y = observation / xaxis=x2

labeljustify=right labelattrs=(weight=bold);
axistable value=Observation &o;
axistable value=DepVar &o;
axistable value=PredictedValue &o;
axistable value=Residual &o;
axistable value=StdErrResidual &o;
axistable value=StudentResidual &o;
axistable value=StdErrMeanPredict &o;



38 F Chapter 3: Axis Tables

axistable value=CooksD &o;
endinnermargin;
highlowplot y=observation

low =eval(min(StudentResidual, 0))
high=eval(max(StudentResidual, 0)) /
xaxis=x2 dataskin=gloss type=bar barwidth=0.5;

referenceline x=0 / xaxis=x2;
endlayout;

endgraph;
end;

quit;

proc sgrender data=r template=StudentizedResidualChart;
label Observation = �Obs� DepVar = �Weight�;
format PredictedValue Residual StdErrResidual StudentResidual

StdErrMeanPredict CooksD 6.2;
run;

The PROC TEMPLATE step creates the graph template, and the PROC SGRENDER step creates the graph.
The HIGHLOWPLOT statement creates the high-low plot. Bars extend from the studentized residual to 0
for negative studentized residuals. Bars extend from 0 to the studentized residual for positive studentized
residuals. A reference line is drawn at X=0, which provides a Y axis for the high-low plot. Most graphs have
an X axis on the bottom of the graph. In this case, the X2 axis is used at the top, so the tick labels and axis
label appear at the top. Additional options control the width and appearance of the bar.

Most of the rest of the template controls the title, axes, and the table. The INNERMARGIN block de�nes the
table. It is aligned in the left of the graphical display, and 5 pixels are inserted after the graph and before the
table (GUTTER=5PX). The inner margin block consists of a series of AXISTABLE statements, one for each
column in the table. The %LET statement after the INNERMARGIN statement creates a macro variable that
contains the options that are common to all of the AXISTABLE statements in order to minimize typing. As
in the HIGHLOWPLOT statement, the X2 axis is used, so the headers for the table appear above the columns
of numbers. Headers are right-justi�ed and appear in a bold font.

The LAYOUT OVERLAY statement controls the axes. The WALLDISPLAY=NONE option removes the
box that surrounds most graphs. Speci�cally, the goal is to suppress vertical lines on the right and left.
The YAXISOPTS= option DISPLAY=NONE suppresses the Y-axis ticks and labels. TYPE=DISCRETE
speci�es the axis type as discrete (the observation number is the Y-axis variable in all the statements), and
REVERSE=TRUE reverses the axis (values decrease as you move up the Y axis). The X2AXISOPTS=
options specify the axis label, a secondary line (the line at the bottom of the graph), a bold font for the axis
label, grid lines, and integer ticks. No X-axis options are speci�ed.

For most of the columns, the variable label (which comes from the header in the PROC REG table) is
displayed above each column. The LABEL statement in the PROC SGRENDER step provides shorter
headers for some of the columns, and the FORMAT statement controls the formats of the columns.

Figure 3.6 Axis Table by Using GTL

You can use the following statements to display the templates that PROC REG uses to make the three different
versions of the residual charts:

proc template;
source Stat.REG.Graphics.StudResCooksDChart;
source Stat.REG.Graphics.StudentResChart;
source Stat.REG.Graphics.CooksDChart;

quit;

The results of this step are not shown. However, the template used in this example contains many of the
elements found in the PROC REG templates. Procedure templates are often complicated because they need
to handle a wide variety of results. In addition, although the procedure developer might have used the macro
language to simplify template development, no macros or macro variables are ever displayed in the template
source. The PROC REG template contains the DESIGNHEIGHT=HEIGHT option in the BEGINGRAPH
statement, whereHeight is a dynamic variable that is set by the procedure. For most graphs, the size of the
graph does not need to vary as a function of the data set size, but it does need to vary for this graph, which
displays one row for each input data set observation. For larger data sets, you could directly specify a height
or set it by using a macro variable. This graph is useful only for small data sets.

You can make a similar plot by using a DATA step and PROC SGPLOT as follows:

data r2;
set r;
low = min(StudentResidual, 0);
high = max(StudentResidual, 0);
format PredictedValue Residual StdErrResidual StudentResidual

StdErrMeanPredict CooksD 6.2;
label Observation = �Obs� DepVar = �Weight� high = �Studentized Residuals�;

run;



40 F Chapter 3: Axis Tables

proc sgplot data=r2 noborder;
title �Residuals�;
yaxistable Observation DepVar PredictedValue Residual StdErrResidual

StudentResidual StdErrMeanPredict CooksD /
y=observation position=left labeljustify=right
labelattrs=(weight=bold);

refline -1 0 1 / axis=x2;
highlow y=observation low=low high=high /

dataskin=gloss type=bar barwidth=0.5 x2axis;
x2axis labelattrs=(weight=bold);
yaxis display=none reverse offsetmin=0 offsetmax=0;

run;

Because PROC SGPLOT does not support expressions, you must use a DATA step to compute the end
points for each bar. You can also set the formats and provide labels in the DATA step and use PROC
SGPLOT to make the graph. The NOBORDER option suppresses the border around the high-low plot. The
YAXISTABLE statement creates the axis table for the named variables. The Y variable isobservation, the
axis table is positioned to the left of the graph, and right-justi�ed and bold headers appear over each column.
The HIGHLOW statement creates the graph; its syntax is similar to the HIGHLOWPLOT statement in the
GTL. The REFLINE statement creates the vertical reference lines. The X2AXIS statement displays a bold
header (axis label) over the graph. The Y axis is reversed, and the DISPLAY=NONE option suppresses the
tick marks and labels (the observation numbers). The zero offsets on the Y axis minimize the space between
the column headers and the body of the graph and table. The results are displayed in Figure 3.7.

Figure 3.7 Axis Table with X Axis Produced by PROC SGPLOT

You can add an X-axis line below the graph as follows:

proc sgplot data=r2 noborder noautolegend;
title �Residuals�;
yaxistable Observation DepVar PredictedValue Residual StdErrResidual

StudentResidual StdErrMeanPredict CooksD /
y=observation position=left labeljustify=right
labelattrs=(weight=bold);

refline -1 0 1 / axis=x2;
scatter y=observation x=low / markerattrs=(size=0);
highlow y=observation low=low high=high /

dataskin=gloss type=bar barwidth=0.5 x2axis;
xaxis display=(nolabel noticks novalues);
x2axis labelattrs=(weight=bold);
yaxis display=none reverse offsetmin=0 offsetmax=0;

run;

The XAXIS statement suppresses labels, ticks, and values so that only a line is displayed. You can specify a
SCATTER statement to add an invisible plot that uses the X axis so that the X-axis line is displayed. The
NOAUTOLEGEND option in the PROC SGPLOT statement suppresses the legend. The results are displayed
in Figure 3.8.

Figure 3.8 Axis Table Produced by PROC SGPLOT

Creating a Forest Plot Using PROC SGPLOT

This example, which is based on an example from Matange and Heath (2011), shows you how to make a
forest plot (see Figure 3.9). Forest plots are used to display the results of a meta-analysis and are typically
used in medical and pharmaceutical research.

Figure 3.9 Forest Plot

This example has three parts, each of which uses PROC SGPLOT:

� “Using an Axis Table to Make a Forest Plot” on page 43. This part shows the easiest way to make
the forest plot: by using the latest capabilities of PROC SGPLOT, which includes the YAXISTABLE
statement.

� “Background: Graphs of Constant Variables” on page 48. This part provides background for the third
part and shows how to display columns of values by plotting constant variables.

� “Using Constant Variables to Make a Forest Plot” on page 54. This part uses constant variables to
make the axis table in a forest plot.

Before SAS 9.4, the only way you could use PROC SGPLOT to make graphs like the forest plot was by using
the constant-variable approach. Both the axis-table method and the constant-variable method are explained in
this example because both illustrate important tools in graph construction. The constant-variable example
also shows other older techniques such as using a SCATTER statement to display error bars rather than using
a HIGHLOW statement. Furthermore, it uses the CNTLIN= data set in PROC FORMAT to generate a format
from a data set.

Using an Axis Table to Make a Forest Plot

This section shows you how to use a YAXISTABLE statement to construct a forest plot. The following step
reads the data set:

data forest;
input Study $1-16 OR LCL UCL Weight;
format weight percent5. OR LCL UCL 5.3;

datalines;
Modano (1967) 0.590 0.096 3.634 1
Modano (1969) 0.429 0.070 2.620 2
Leighton (1972) 0.394 0.076 2.055 2
Borodan (1981) 0.464 0.201 1.074 3.5
Novak (1992) 0.490 0.088 2.737 2
Stawer (1998) 1.250 0.479 3.261 3
Adams (1999) 0.143 0.082 0.250 4
Soloway (2000) 0.718 0.237 2.179 3
Truark (2002) 0.129 0.027 0.605 2.5
Fayney (2005) 0.313 0.054 1.805 2
Modano (2006) 0.429 0.070 2.620 2
Adams (2007) 0.143 0.082 0.250 4
Soloway (2008) 0.718 0.237 2.179 3
Truark (2010) 0.129 0.027 0.605 2.5
Fayney (2012) 0.313 0.054 1.805 2
Overall 0.328 0.233 0.462 .
;

The variableOR contains the odds ratio,LCL contains the lower con�dence limit,UCL contains the upper
con�dence limit, andWeight contains the study weight. The FORMAT statement assigns formats to the
speci�ed variables.

You will need some additional variables to display the study names on the Y axis. The following steps create
and display these and other new variables (which are explained after the data set is displayed in Figure 3.10):

data forest;
set forest nobs=nobs;
if _n_ eq 1 then call symputx(�nobs�, nobs);
nAll = nobs + 1 - _n_; * 1-16 for everything;
nStudy = ifn(study eq �Overall�, . , nall); * 2-16 for studies;
nOverall = ifn(study eq �Overall�, nall, . ); * 1 for overall;
if n(weight) then do;

weight = weight * 0.05; / * Rescale weights * /
lo = OR / (10 ** (weight/2)); / * Bar width * /
hi = OR * (10 ** (weight/2)); / * shows study�s weight * /

end;
run;

proc print noobs;
run;

The �rst IFN function returns a missing value whenstudy = �Overall� and otherwise returns the value of
the variablenAll. The results are displayed in Figure 3.10.

Figure 3.10 Forest Plot Format Data

The data set has three Y-axis variables (nAll, nStudy, andnOverall), each of which contains integers that
indicate the number for the relevant study names. ThenStudy (15 nonmissing values) andnOverall (one
nonmissing value) Y-axis variables are needed because the individual study results are displayed differently
from the overall results. Hence, separate plotting statements are needed, and each needs a different Y-axis
variable. ThenAll variable (no missing values) is used to create the other two variables. The widths of the
bars show the weights of the studies. Widths are computed on an exponential scale and are later displayed on
a log scale.

The next step creates the forest plot:

title �Impact of Treatment on Mortality by Study�;
title2 h=8pt �Odds Ratio and 95% CL�;

proc sgplot data=forest noautolegend nocycleattrs;
refline 1 100 / axis=x;
refline 0.1 10 / axis=x lineattrs=(pattern=shortdash) transparency=0.5;
highlow y=nStudy low=LCL high=UCL / lowcap=serif highcap=serif

lineattrs=graphdata1;
scatter y=nOverall x=OR / markerattrs=graphdata2

(symbol=diamondfilled size=10);
highlow y=nStudy low=lo high=hi / type=bar intervalbarwidth=.1in

nooutline fillattrs=graphdata1;
yaxistable study / y=nAll position=left location=outside

nolabel valuejustify=right pad=(right=0px);
yaxistable OR LCL UCL weight / y=nStudy position=right location=inside

labelhalign=center labelattrs=(weight=bold)
pad=(left=20px right=10px);

inset � Favors Treatment� / position=bottomleft;
inset �Favors Placebo � / position=bottomright;
xaxis type=log min=0.01 max=100 minor display=(nolabel)

offsetmin=0 offsetmax=0;
yaxis display=none offsetmin=0.1 offsetmax=0.05 values=(1 to &nobs);

run;

The NOCYCLEATTRS option in the PROC SGPLOT statement speci�es that PROC SGPLOT should not
automatically create unique attributes (such as colors) for the elements in the graph. All the attributes are
speci�ed in the options. The X axis ranges from MIN=0.01 to MAX=100 and has solid references lines at 1
and 100 and dashed reference lines at 0.1 and 10. The �rst HIGHLOW statement displays the individual
study results as error bars. The SCATTER statement displays the overall results as a �lled diamond. The
second HIGHLOW statement displays the boxes that show the study weights. The �rst YAXISTABLE
statement displays the study names on the Y axis, outside the graph and to its left. The second YAXISTABLE
statement displays the odds ratio, con�dence limits, and weight. The inset strings�Favors Treatment�
and�Favors Placebo� label the two parts of the X axis. Labels are placed on the bottom left and bottom
right of the graph. (The OFFSETMIN=0.1 option in the YAXIS statement reserves space for those two
strings.) In the XAXIS statement, the TYPE=LOG option speci�es a log scale and the OFFSETMIN=0 and
OFFSETMAX=0 options ensure that there is no extra space beyond the minimum and maximum values (0.01
and 100, respectively). The YAXIS statement speci�es the values 1 to 16 (1 to the value of the macro variable
&nobs) on the Y axis. However, actual study names are displayed instead of the integers. The results are
displayed in Figure 3.11.

Figure 3.11 Forest Plot Produced by Two YAXISTABLE Statements

The second YAXISTABLE statement creates an axis table that has four columns and is positioned inside the
graph and to the right. The Y-axis variable isnStudy, and headers are centered and bold. Padding controls
the space to the left and right of the table. The axis table in the graph in Figure 3.11 could be improved. The
following step adjusts the columns so that they are more evenly spaced:

proc sgplot data=forest noautolegend nocycleattrs;
refline 1 100 / axis=x;
refline 0.1 10 / axis=x lineattrs=(pattern=shortdash) transparency=0.5;
highlow y=nStudy low=LCL high=UCL / lowcap=serif highcap=serif

lineattrs=graphdata1;
scatter y=nOverall x=OR / markerattrs=graphdata2

(symbol=diamondfilled size=10);
highlow y=nStudy low=lo high=hi / type=bar intervalbarwidth=.1in

nooutline fillattrs=graphdata1;
yaxistable study / y=nAll position=left location=outside

nolabel valuejustify=right pad=(right=0px);
yaxistable OR LCL UCL / y=nStudy position=right location=inside

labelhalign=center labelattrs=(weight=bold)
pad=(left=20px right=0px);

yaxistable weight / y=nStudy position=right location=inside
labelhalign=center labelattrs=(weight=bold)
pad=(left=10px right=10px) valuehalign=center;

inset � Favors Treatment� / position=bottomleft;
inset �Favors Placebo � / position=bottomright;
xaxis type=log min=0.01 max=100 minor display=(nolabel)

offsetmin=0 offsetmax=0;
yaxis display=none offsetmin=0.1 offsetmax=0.05 values=(1 to &nobs);

run;

The results are displayed in Figure 3.12.

Figure 3.12 Forest Plot Produced by Three YAXISTABLE Statements

The default spacing of the columns depends on the widths of the values and column headers. The �rst three
columns (OR, LCL, andUCL) have a uniform width and headers that are less wide than the values, but the
Weight variable is different. The graph in Figure 3.12 is created by separating theWeight variable from the
OR, LCL, andUCL variables, and specifying it in a separate YAXISTABLE statement in order to individually
control its position. The YAXISTABLE statement for theWeight variable centers the values relative to the
header (but maintains the right-justi�cation relative to each other). All the YAXISTABLE statements control
the padding to provide more uniform spacing between the columns.

You can achieve a different look by specifying LOCATION=OUTSIDE in all YAXISTABLE statements to
move the axis table outside the graph as follows:

proc sgplot data=forest noautolegend nocycleattrs;
refline 1 / axis=x;
refline 0.1 10 / axis=x lineattrs=(pattern=shortdash) transparency=0.5;
highlow y=nStudy low=LCL high=UCL / lowcap=serif highcap=serif

lineattrs=graphdata1;
scatter y=nOverall x=OR / markerattrs=graphdata2

(symbol=diamondfilled size=10);
highlow y=nStudy low=lo high=hi / type=bar intervalbarwidth=.1in

nooutline fillattrs=graphdata1;
yaxistable study / y=nAll position=left location=outside

nolabel valuejustify=right pad=(right=0px);
yaxistable OR LCL UCL / y=nStudy position=right location=outside

labelhalign=center labelattrs=(weight=bold)
pad=(left=20px right=0px);

yaxistable weight / y=nStudy position=right location=outside
labelhalign=center labelattrs=(weight=bold)
pad=(left=10px right=10px) valuehalign=center;

inset � Favors Treatment� / position=bottomleft;
inset �Favors Placebo � / position=bottomright;
xaxis type=log min=0.01 max=100 minor display=(nolabel)

offsetmin=0 offsetmax=0;
yaxis display=none offsetmin=0.1 offsetmax=0.05 values=(1 to &nobs);

run;

The results are displayed in Figure 3.13.

Figure 3.13 Forest Plot with an Outside Axis Table

Background: Graphs of Constant Variables

The rest of this example shows an alternative way to make the forest plot that involves constant variables and
offsets. This section illustrates those concepts by using simple arti�cial data sets. Section “Using Constant
Variables to Make a Forest Plot” on page 54 shows you how to construct the forest plot. Also see the section
“Multiple Axes and Highlighted Points” on page 13 for an introduction to multiple axes.

The following steps create and display a SAS data set that will be helpful in understanding this method of
constructing forest plots:

data x;
retain x1 1 x2 2 x3 3 c1 �A� c2 �B� c3 �C�;
do y = 1 to 10;

l1 = substr(�ABCDEFGHIJ�, y, 1.);
l2 = put(y, words12.);
l3 = y;
output;

end;
run;

proc print noobs;
run;

The data are displayed in Figure 3.14.

Figure 3.14 Arti�cial Data

The data consist of three constant numeric variables (x1–x3), three constant character variables (c1–c3), ay
variable that contains the observation number, and three label variables (l1–l3). Plotting constant variables is
central to this method of constructing the forest plot. The RETAIN statement retains values across multiple
passes through the DATA step rather than initializing them to missing each time. This is a one-pass DATA
step (so retaining is not required), but the RETAIN statement provides a convenient syntax for creating and
initializing constant variables.

The following step creates a graph that has the variabley on the Y axis and the constant numeric variables
x1–x3 on the X axis:

proc sgplot noautolegend;
scatter x=x1 y=y;
scatter x=x2 y=y;
scatter x=x3 y=y;

run;

The results are displayed in Figure 3.15. The X axis is linearly scaled, and all threex variables share the
same X axis.

Figure 3.15 Plotting Constant Numeric Variables

The following step creates a graph that has the variabley on the Y axis and the constant character variables
c1–c3 on the X axis.

proc sgplot noautolegend;
scatter x=c1 y=y;
scatter x=c2 y=y;
scatter x=c3 y=y;

run;

The results are displayed in Figure 3.16. The X axis is discrete, and values appear in the order in which they
are encountered.

Figure 3.16 Plotting Constant Character Variables

The following step overlays Figure 3.15 and Figure 3.16 and moves the character variables to the X2 axis:

proc sgplot noautolegend;
scatter x=x1 y=y;
scatter x=x2 y=y;
scatter x=x3 y=y;
scatter x=c1 y=y / x2axis markerattrs=(symbol=square size=3px);
scatter x=c2 y=y / x2axis markerattrs=(symbol=square size=3px);
scatter x=c3 y=y / x2axis markerattrs=(symbol=square size=3px);

run;

The results are displayed in Figure 3.17. It might not be obvious yet, but this is an important intermediate
step in understanding how to construct the forest plot.



Creating a Forest Plot Using PROC SGPLOT F 51

Figure 3.17 Using the X and X2 Axis

Next, you need offsets. Offsets are commonly used to move the points in from the axes. The following step
creates Figure 3.18:

proc sgplot noautolegend;
scatter x=x1 y=y;
scatter x=x2 y=y;
scatter x=x3 y=y;
xaxis offsetmin=0.1 offsetmax=0.1;
yaxis offsetmin=0.1 offsetmax=0.1;

run;

Figure 3.18 Standard Offsets

Compare Figure 3.18 and Figure 3.15 and notice how Figure 3.18 has extra space at the minimum and
maximum end of both the X and Y axes. The 0.1 offsets move the points in from the axes by a distance that
is 10% of the axis length.

You can create side-by-side plots by using multiple axes and much larger offsets. The following step creates
Figure 3.19:

proc sgplot noautolegend;
scatter x=x1 y=y;
scatter x=x2 y=y;
scatter x=x3 y=y;
scatter x=c1 y=y / x2axis markerattrs=(symbol=square size=3px);
scatter x=c2 y=y / x2axis markerattrs=(symbol=square size=3px);
scatter x=c3 y=y / x2axis markerattrs=(symbol=square size=3px);
xaxis offsetmax=0.6;
x2axis offsetmin=0.6;

run;

Figure 3.19 Two Sets of Plots with Each Set Offset

This graph is starting to resemble a forest plot. There is a graph on the left, a numeric axis on the bottom, tick
labels on the top that are starting to resemble column headers, and a graph on the right that will become the
table. The OFFSETMAX=0.6 option in the XAXIS statement reserves 60% of the graph space for the graph
on the right. The OFFSETMIN=0.6 option in the X2AXIS statement reserves 60% of the graph space for the
graph on the left.

The next step, which creates Figure 3.20, is the �nal step before you learn how to construct a forest plot by
using the constant-variable approach:

proc sgplot noautolegend;
scatter x=x1 y=y;
scatter x=x2 y=y;
scatter x=x3 y=y;
scatter x=c1 y=y / x2axis markerchar=l1;
scatter x=c2 y=y / x2axis markerchar=l2;
scatter x=c3 y=y / x2axis markerchar=l3;
xaxis offsetmax=0.6;
x2axis offsetmin=0.6 display=(noticks nolabel);

run;

Figure 3.20 A Graph and a Table Displayed Together

In this step, the MARKERCHAR= option is used to display the label variables in the right part of the graph.
The DISPLAY=(NOTICKS NOLABEL) option displays only the constant character variable values on the
X2 axis, which now look like column headers, completing the transformation from a graph to a table. The
constant variables are required because all the values in each table column have a constant X2 coordinate.

Using Constant Variables to Make a Forest Plot

The next step creates the data set that is used to make the forest plot:

data forest;
input Study $1-16 OddsRatio LowerCL UpperCL Weight;
format weight percent5. oddsratio lowercl uppercl 5.3;
retain OR �OR� LCL �LCL� UCL �UCL� WT �Weight� FmtName �Study�;
if n(weight) then weight = weight * 0.05;

datalines;
Modano (1967) 0.590 0.096 3.634 1
Modano (1969) 0.429 0.070 2.620 2
Leighton (1972) 0.394 0.076 2.055 2
Borodan (1981) 0.464 0.201 1.074 3.5
Novak (1992) 0.490 0.088 2.737 2
Stawer (1998) 1.250 0.479 3.261 3
Adams (1999) 0.143 0.082 0.250 4
Soloway (2000) 0.718 0.237 2.179 3
Truark (2002) 0.129 0.027 0.605 2.5
Fayney (2005) 0.313 0.054 1.805 2
Modano (2006) 0.429 0.070 2.620 2
Adams (2007) 0.143 0.082 0.250 4
Soloway (2008) 0.718 0.237 2.179 3
Truark (2010) 0.129 0.027 0.605 2.5
Fayney (2012) 0.313 0.054 1.805 2
Overall 0.328 0.233 0.462 .
;

The RETAIN statement creates constant character variables that provide the positions and column headers
for the table. The variableOR (odds ratio) has the constant value�OR� , LCL (lower con�dence limit) has the
constant value�LCL� , UCL (upper con�dence limit) has the constant value�UCL� , andWt has the constant
value�Weight� .

The constant variableFmtName is used in a subsequent step to create a SAS format. You will use the format
to display the study names on the Y axis while specifying two integer variables in the plotting statements.
The following steps create and display these variables, which are shown in Figure 3.21:

data forest2; / * Make format input data and Y-axis variables * /
set forest nobs=nobs;
Start = nobs + 1 - _n_; * Need 1-16 for format;
nStudy = ifn(study eq �Overall�, . , start); * Need 2-16 for studies;
nOverall = ifn(study eq �Overall�, start, . ); * Need 1 for overall;

run;

proc print noobs;
run;

Figure 3.21 Forest Plot Data

In the next step, PROC FORMAT creates the format from an input SAS data set. It needs a variable called
Label, which contains the values that are displayed. The variable can be created by renaming the variable
Study. PROC FORMAT also needs a variable calledFmtName, which contains the name of the format
that you want to create (Study). Finally, PROC FORMAT needs a variable calledStart, which contains the
input values (1–16) that are replaced by the values of theLabel variable. The new format is applied to the
variablesnStudy andnOverall, which will both be speci�ed on the Y axis. Two study name variables are
needed because the overall results are displayed differently from the individual study results. Hence, separate
plotting statements are needed for the studies and for the overall results, each of which uses separate Y-axis
variables. The following step runs PROC FORMAT and creates the format STUDY, which maps the integers
1–16 to the study names:

proc format library=work cntlin=forest2(keep=fmtname study start
rename=(study=label));

run;

The next step applies the format and creates a new data set that has two more variables that are needed (lo
andhi):

data forest3;
set forest2(drop=fmtname start) nobs=nobs;

/ * Compute the width of the box proportional to weight on log axis. * /
if n(weight) then lo = oddsratio / (10 ** (weight/2)); / * Bar width * /
if n(weight) then hi = oddsratio * (10 ** (weight/2)); / * shows study�s * /

/ * weight. * /
if _n_ eq 1 then call symputx(�nobs�, nobs);
format nStudy nOverall study.;

run;

The widths of the bars show the weights of the studies. Widths are computed on an exponential scale and are
later displayed on a log scale.

The next series of steps builds up to the forest plot by �rst constructing the shell of the plot and then adding
to it. This step creates Figure 3.22:

title �Impact of Treatment on Mortality by Study�;
title2 h=8pt �Odds Ratio and 95% CL�;

proc sgplot data=forest3 noautolegend nocycleattrs;
refline 1 100 / axis=x;
refline 0.1 10 / axis=x lineattrs=(pattern=shortdash) transparency=0.5;
scatter y=nOverall x=oddsratio / markerattrs=graphdata2

(symbol=diamondfilled size=10);
inset � Favors Treatment� / position=bottomleft;
inset �Favors Placebo� / position=bottom;
xaxis type=log offsetmin=0 offsetmax=0.35 min=0.01 max=100

minor display=(nolabel);
x2axis offsetmin=0.7 display=(noticks nolabel);
yaxis display=(noticks nolabel) offsetmin=0.1 offsetmax=0.05

values=(1 to &nobs);
run;

The SCATTER statement produces the display of the overall results. All the other statements control titles,
reference lines, inset text, and the axes. The NOCYCLEATTRS option in the PROC SGPLOT statement
speci�es that PROC SGPLOT should not automatically create unique attributes for the elements in the graph.
All the attributes are speci�ed in the options. The OFFSETMAX=0.35 option in the XAXIS statement
reserves 35% of the space to the right of the graph for the table. The OFFSETMIN=0.7 option in the X2AXIS
statement reserves 70% of the space to the left of the table for the graph. The YAXIS statement speci�es the
values 1 to 16 (1 to the value of the macro variable&nobs) on the Y axis. Because the Y-axis variables have
formats associated with them, the study names are displayed instead of integers.

